
Le Journal du Soft ( 4 )
où la ROMMIRE vous en fera voir de toutes les couleurs

par André Chéramy et Claude Sittler

Nous allons décrire une ROM datée 1985 qui fût mise
au point par François Sittler pour afficher des mires. Il s'agit
d'une application vidéo pour radioamateurs: l'Oric est devenu un
générateur de mires. Les images générées étaient récupérées sur
la sortie vidéo de l'Oric, mais c'est un problème que nous
traiterons une autre fois. 

Outre le plaisir de tripatouiller la ROM de l'Atmos
avec nous et celui de voir votre récompense sur l'écran, cet
exercice vous donnera un exemple pour remplacer les
commandes que vous n'utilisez pas (ici EDIT, STORE,
RECALL et LORES) par vos commandes personnelles. A vous
ensuite de feuilleter vos vieux Théoric (voir notamment les
articles de Thierry Legal) ou de mettre au point vos propres
programmes et de les insérer dans une ROM. Faites-nous part
de vos essais.

Pour suivre pas à pas nos indications, vous avez besoin

de peu de choses. Un fichier ATMOS.ROM (voir "J. Soft n°1)
et votre moniteur habituel. Puis testez votre travail avec
OVERLAY.COM ou avec Euphoric (voir "L'Oric en Folie").

Bootez votre ordinateur, tapez HIMEM#FFF5
ATMOS.ROM,A#10005 et MONAC15 (ou votre moniteur
favori). Pour chacune des modifications indiquées ci-après,
listez ou dumpez le code existant, tapez vos modifications,
vérifiez en re-listant. A la fin, sauvez votre travail avec
SAVE"ROMMIRE", A#1000,E#4FFF5 suivi d'un
STATUS"ROMMIRE",A#3500 si vous avez l'intention de
s h o o t e r  l e  f i c h i e r  d a n s  u n e  E P R O M  o u
STATUS"ROMMIRE",A#2000 si vous voulez le tester avec
OVERLAY.COM. Pour l'utilisation avec Euphoric, faites un
CSAVE"ROMMIRE.K7",A#1000,E#4FFF5 puis re-sauvez
sous le nom ROMMIRE.ROM après avoir retiré l'entête K7
avec un éditeur hexadécimal.

1) MODIFICATION DES ADRESSES D'EXÉCUTION DES COMMANDES
En C008/C009 (1008/1009), remplacez 91 C6 (adresse exéc de EDIT) par 86 E9 (E987 adresse exéc de MIRE)
En C00A/C00B (100A/100B), remplacez 86 E9 (adresse exéc de STORE) par 16 F8 (F817 adresse exéc de DEGRA)
En C00C/C00D (100C/100D), remplacez D0 E9 (adresse exéc de RECALL) par CE E9 (E9CF adresse exéc de DEFINI)
En C018/C019 (1018/1019), remplacez DD D9 (adresse exéc de LORES) par 2A EA (EA2B adresse exéc de CONVE)

2) MODIFICATION DU NOM DES COMMANDES
De C0ED à C0F0 (10ED à 10F0), remplacez EDIT (45 44 49 D4) par MIRE (4D 49 52 C5)
De C0F1 à C0F5 (10F1 à 10F5), remplacez STORE (53 54 4F 52 C5) par DEGRA (44 45 47 52 C1)
De C0F6 à C0FB (10F6 à 10FB), remplacez RECALL (52 45 43 41 4C CC ) par DEFINI (44 45 46 49 4E C9)
De C110 à C114 (1110 à 1114), remplacez LORES (4C 4F 52 45 D3) par CONVE (43 4F 4E 56 C5)

3) MODIFICATION DU COPYRIGHT MICROSOFT
De E435 à E43E (3435 à 343E), remplacez A1 54 46 8F 13 8F 52 43 89 CD (Microsoft, codé)

par A1 10 14 11 15 12 16 13 17 CD (table des attributs de couleur PAPER)

4) NOUVELLE COMMANDE MIRE  située dans le code de STORE à partir de E987, c'est à dire 3987 en RAM pour nous
3987 A2 BB LDX #BB place l'adresse BB80
3989 A0 80 LDY #80 du début de l'écran 
398B 86 01 STX 01 (ligne service)
398D 84 00 STY 00 en 00/01, puis prépare 
398F A0 A8 LDY #A8 l'adresse BBA8 (ligne suivante)
3991 86 03 STX 03 et la place 
3993 84 02 STY 02 en 02/03, il faudra 
3995 A2 0E LDX #0E dessiner 14 rangées de carrés
3997 A0 00 LDY #00 index pour écriture ultérieure
3999 AD 08 03 LDA 0308 VIAT2L valeur LL du Timer 2
399C EA NOP (nombre au hasard)
399D 29 07 AND #07 force à zéro les bits b7 à b3
399F 09 10 ORA #10 force à 1 le b4, le résultat va de
#10 (= 16) à #17 (= 23), attributs de papier noir à papier blanc
39A1 91 00 STA (00),Y écrit cet attribut au 1er pointeur
39A3 91 02 STA (02),Y et idem sur la ligne suivante
39A5 C8 INY pour indexer la position suivante
39A6 91 00 STA (00),Y écrit même attribut à droite du 1er

39A8 91 02 STA (02),Y et idem sur la ligne suivante
on a donc un carré au format 2x2

39AA C8 INY pour indexer la position suivante
39AB C0 28 CPY #28 la ligne est-elle finie (#28 = 40)
39AD D0 EA BNE 3999 sinon, on reboucle
39AF A5 00 LDA 00 lorsque la marge de droite est 
39B1 18 CLC atteinte, on effectue les
39B2 69 50 ADC #50 opérations  suivantes
39B4 85 00 STA 00 adresse en 00/01 = adresse + #50
39B6 A5 01 LDA 01 (= 80 soit deux lignes)
39B8 69 00 ADC #00
39BA 85 01 STA 01
39BC A5 02 LDA 02
39BE 18 CLC 
39BF 69 50 ADC #50
39C1 85 02 STA 02 idem adresse en 00/01 =
39C3 A5 03 LDA 03 adresse + #50
39C5 69 00 ADC #00
39C7 85 03 STA 03 enfin, décrémente le nombre
39C9 CA DEX de rangées de carrés à dessiner
39CA D0 CB BNE 3997 s'il en reste, reboucle
39CC 4C 9A EA JMP EA9A sinon, attente touche clavier

5) NOUVELLE COMMANDE DEFINI située dans l'ancien code de STORE/RECALL à partir de E9CF, c'est à dire 39CF en RAM
39CF 20 AB EA JSR EAAB mise à 0 des paramètres graphiques de 02E0 à 02E6 et passage en HIRES
a) Trace 5 lignes verticales espacées de 40 pixels
39D2 A2 28 LDX #28 40 = coord x de 1e ligne verticale
39D4 86 33 STX 33 sauve la valeur courante de x
39D6 A6 33 LDX 33 relit la valeur courante de x

39D8 A0 00 LDY #00 positionne Y en haut de l'écran
39DA 20 63 EA JSR EA63 CURSET (X=x, Y=y et FB=1)
39DD A2 00 LDX #00 force la valeur de ∆x à zéro
39DF A0 C7 LDY #C7 force la valeur de ∆y à 199



39E1 20 71 EA JSR EA71 DRAW (X=∆x, Y=∆y et FB=1)
39E4 20 7F EA JSR EA7F x = x + 40 (valeur située en 33)

39E7 C9 F0 CMP #F0 x atteint-il 240 (le bord droit)?
39E9 D0 EB BNE 39D6 sinon, reboucle

b) Trace 4 lignes horizontales espacées de 40 pixels
39EB A9 28 LDA #28 40 = coord y de 1e ligne horizont
39ED 85 33 STA 33 sauve la valeur courante de y
39EF A4 33 LDY 33 relit la valeur courante de y
39F1 A2 00 LDX #00 positionne x à gauche de l'écran
39F3 20 63 EA JSR EA63 CURSET (X=x, Y=y et FB=1)

39F6 A2 EF LDX #EF force la valeur de ∆x à 239
39F8 A0 00 LDY #00 force la valeur de ∆y à zéro
39FA 20 71 EA JSR EA71 DRAW (X=∆x, Y=∆y et FB=1)
39FD 20 7F EA JSR EA7F y = y + 40 (valeur située en 33)
3A00 C9 C8 CMP #C8   y atteint-il 200 (le bord inférieur)?
3A02 D0 EB BNE 39EF sinon, reboucle

c) Trace un point au centre de chaque carré délimité par les lignes
3A04 A0 14 LDY #14 20 = coordonnée y du 1er point
3A06 84 33 STY 33 sauve la valeur courante de y
3A08 A2 14 LDX #14 20 = coordonnée x du 1er point
3A0A 86 34 STX 34 sauve la valeur courante de x
3A0C A6 34 LDX 34 relit la valeur courante de x
3A0E A4 33 LDY 33 relit la valeur courante de y
3A10 20 63 EA JSR EA63 CURSET (X=x, Y=y et FB=1)
3A13 20 8F EA JSR EA8F CICLE rayon 1 pixel au curseur
3A16 A5 34 LDA 34 mise à jour x = x+#28 c'est à dire

3A18 18 CLC + 40 (coord du point suivant)
3A19 69 28 ADC #28 qui prend donc les valeurs #14,
3A1B 85 34 STA 34 #3C, #64, #8C, #B4, #DC et #104
3A1D C9 04 CMP #04 #104 = 260 dépasse du bord droit
3A1F D0 EB BNE 3A0C reboucle tant que pas le cas
3A21 20 7F EA JSR EA7F y = y + 40 (valeur située en 33)
3A24 C9 DC CMP #DC #DC=220 dépasse du bord du bas
3A26 D0 E0 BNE 3A08 reboucle si ce n'est pas le cas
3A28 4C 9A EA JMP EA9A si fini, attente touche clavier

6) NOUVELLE COMMANDE CONVE située dans l'ancien code de RECALL à partir de EA2B, c'est à dire 3A2B pour nous
3A2B 20 AB EA JSR EAAB mise à 0 des paramètres graphiques de 02E0 à 02E6 et passage en HIRES
a) Trace 23 lignes verticales espacées de 10 pixels
3A2E A2 0A LDX #0A 10 = coord x de 1e ligne verticale
3A30 86 33 STX 33 sauve la valeur courante de x
3A32 A6 33 LDX 33 relit la valeur courante de x
3A34 A0 00 LDY #00 place le curseur en haut de l'écran
3A36 20 63 EA JSR EA63 CURSET (X=x, Y=y et FB=1)

3A39 A2 00 LDX #00 force la valeur de ∆x à zéro
3A3B A0 C7 LDY #C7 force la valeur de ∆y à 199
3A3D 20 71 EA JSR EA71 DRAW (X=∆x, Y=∆y et FB=1)
3A40 20 87 EA JSR EA87 x = x + 10 (valeur située en 33)
3A43 C9 F0 CMP #F0 x atteint-il 240 (le bord droit)?
3A45 D0 EB BNE 3A32 sinon, reboucle

b) Trace 4 lignes horizontales espacées de 10 pixels
3A47 A2 0A LDX #0A 10 = coord y de 1e ligne horizont
3A49 86 33 STX 33 sauve la valeur courante de y
3A4B A2 00 LDX #00 positionne x à gauche de l'écran
3A4D A4 33 LDY 33 relit la valeur courante de y
3A4F 20 63 EA JSR EA63 CURSET (X=x, Y=y et FB=1)
3A52 A2 EF LDX #EF force la valeur de ∆x à 239

3A54 A0 00 LDY #00 force la valeur de ∆y à zéro
3A56 20 71 EA JSR EA71 DRAW (X=∆x, Y=∆y et FB=1)
3A59 20 87 EA JSR EA87 y = y + 10 (valeur située en 33)
3A5C C9 C8 CMP #C8 y atteint-il 200 ( bord inférieur)?
3A5E D0 EB BNE 3A4B sinon, reboucle
3A60 4C 9A EA JMP EA9A si fini, attente touche clavier

7) SÉRIE DE SOUS-PROGRAMME APPELÉS PAR DEFINI, CONVE etc, situés dans le code de RECALL
a) Sous-progr EA63, exécute CURSET selon X=x, Y=y et FB=1
3A63 8E E1 02 STX 02E1 coordonnée x
3A66 8C E3 02 STY 02E3 coordonnée y

3A69 A9 01 LDA #01 valeur de FB (#01 =
3A6B 8D E5 02 STA 02E5 couleur de INK)
3A6E 4C C8 F0 JMP F0C8 exécute cde CURSET et retourne

b) Sous-progr EA71, exécute DRAW selon X=∆x, Y=∆y et FB=1
3A71 8E E1 02 STX 02E1 valeur de ∆x
3A74 8C E3 02 STY 02E3 valeur de ∆y

3A77 A9 01 LDA #01 valeur de FB (#01 = 
3A79 8D E5 02 STA 02E5 couleur de INK)
3A7C 4C 10 F1 JMP F110 exécute cde DRAW et retourne

c) Sous-progr EA7F, ajoute 40 pixels à la variable en 33
3A7F A5 33 LDA 33 relit  la valeur courante de
3A81 18 CLC  x ou y stockée en 33

3A82 69 28 ADC #28 lui ajoute 40
3A84 85 33 STA 33 la remet en place
3A86 60 RTS            retourne avec nouv valeur dans A

d) Sous-progr EA87, ajoute 40 pixels à la variable en 33
3A87 A5 33 LDA 33 relit la valeur courante de
3A89 18 CLC  x ou y stockée en 33

3A8A 69 0A ADC #0A lui ajoute 10
3A8C 85 33 STA 33 la remet en place
3A8E 60 RTS            retourne avec nouv valeur dans A

e) Sous-progr EA8F, trace un cercle de rayon 1 pixel au curseur: avec un tel rayon, le cercle est un carré de 2x2 !
3A8F A9 01 LDA #01 1 pixel
3A91 8D E1 02 STA 02E1 valeur du rayon

3A94 8D E3 02 STA 02E3 valeur FB
3A97 4C 7F F3 JMP F37F CIRCLE au curseur et retourne

f) Sous-progr EA9A, attend une pression de touche et retourne au Ready
3A9A AD 6A 02 LDA 026A registre des modes console sauvé
3A9D 48 PHA sur la pile; force curseur caché,
3A9E A9 0A LDA #0A vidéo active, bruit clavier OFF
3AA0 8D 6A 02 STA 026A nouveaux paramètres console

3AA3 20 E8 C5 JSR C5E8 routine attendre touche clavier
3AA6 68 PLA récupérer les paramètres d'origine
3AA7 8D 6A 02 STA 026A et les remettre en place
3AAA 60 RTS retour au Ready

g) Sous-progr EAAB, mise à zéro des paramètres graphiques et passage en HIRES
3AAB A2 07 LDX #07 pour mettre à zéro les 7 octets
3AAD A9 00 LDA #00 de 02E0 à 02E6

3AAF 9D DF 02 STA 02DF,X soit 02E0 pour le drapeau
d'erreur, 02E1/02E2 pour la valeur de x/∆x, 02E3/02E4 pour



DEFINI CONVE

MIRE DEGRA

celle de y/∆y et 02E5/02E6 pour la valeur de FB
3AB2 CA DEX

3AB3 D0 FA BNE 3AAF rebouclage tant qu'il en reste 
3AB5 4C 33 EC JMP EC33 commande HIRES

8) MODIFICATION DU COPYRIGHT Lors du boot, il est important de pouvoir identifier à quelle ROM on a affaire.
Dans la ROM que nous avons testée, les modifications suivantes avaient été apportées:
De ED96 à EDEF, le message "ORIC EXTENDED BASIC V1.1" et "© 1983 TANGERINE"
soit  4F 52 49 43 20 45 58 54  45 4E 44 45 44 20 42 41 53 49 43 20 56 31 2E 31 0D 0A
et 60 20 31 39 38 33 20 54 41 4E 47 45 52 49 4E 45 0D 0A
est devenu " ORIC MINI MIRE S.F SYST." et "© 1985 TANGERINE"
soit 4F 52 49 43 20 4D 49 4E 49 20 4D 49 52 45 20 53 2E 46 20 53 59 53 54 2E 0D 0A
et 60 20 31 39 38 35 20 54 41 4E 47 45 52 49 4E 45 0D 0A

9) ANNULATION DE LA ROUTINE "GÉNÉRER LES CARACTÈRES ALTERNÉS" 
Qui se trouvait en F816 et sera remplacée par la commande DEGRA
4816 60 RTS la compatibilité est assurée: les appels à cette routine ne planteront pas la machine !

10) NOUVELLE COMMANDE DEGRA située à la place du code de génération des caractères alternés à partir de F817
4817 20 CE CC JSR CCCE effectue un CLS, puis
481A A2 27 LDX #27 efface la ligne service
481C A9 20 LDA #20
481E 9D 80 BB STA BB80,X en y écrivant 40 espaces
4821 CA DEX 
4822 10 F8 BPL 481C reboucle tant qu'il en reste
4824 A2 BB LDX #BB place l'adresse BB80
4826 A0 80 LDY #80 du début de l'écran 
4828 86 01 STX 01 (ligne service)
482A 84 00 STY 00 en 0000/0001
482C A9 1C LDA #1C index pour 28 lignes
482E 85 02 STA 02 stocké en 0002
4830 A0 00 LDY #00 index pour écriture dans l'écran
4832 A2 00 LDX #00 index pour lecture dans la table
4834 BD 36 E4 LDA E436,X lit attribut PAPER
4837 91 00 STA (00),Y l'écrit dans l'écran
4839 98 TYA 
483A 18 CLC 
483B 69 05 ADC #05 effectue Y = Y + 5
483D A8 TAY

483E E8 INX vise attribut suivant
483F C0 28 CPY #28 fin de ligne atteinte ?
4841 D0 F1 BNE 4834 sinon, reboucle
4843 A5 00 LDA 00 si oui, mise à jour
4845 18 CLC 
4846 69 28 ADC #28 00/01 = 00/01+ 40
4848 85 00 STA 00
484A A5 01 LDA 01 pour viser la ligne suivante
484C 69 00 ADC #00
484E 85 01 STA 01
4850 C6 02 DEC 02 décrémente nombre de lignes
4852 D0 DC BNE 4830 reboucle tant qu'il en reste
4854 AD 6A 02 LDA 026A registre des modes console
4857 48 PHA  sauvé sur la pile
4858 A9 0A LDA #0A 0000 1010 soit curseur caché,
485A 8D 6A 02 STA 026A vidéo active, bruit clavier OFF
485D 20 E8 C5 JSR C5E8 attend une touche au clavier
4860 68 PLA  récupérer paramètres d'origine
4861 8D 6A 02 STA 026A et les remettre en place
4864 60 RTS retour au Ready

11) ANNULATION DE L'APPEL À LA ROUTINE "GÉNÉRER LES CARACTÈRES ALTERNÉS" situé en F8D5
De 48D5 à 48D7, remplacer JSR F816 (20 16 F8) par 3 NOPs (EA EA EA).


